1 Eigenvalues and Singular Values of a Symmetric Matrix

Since $A = A^T$, A has an orthonormal set of vectors and

$$A = Q\Lambda Q^T$$

where Q is orthogonal ($Q^{-1} = Q^T$) and

$$\Lambda = \text{diag}(\lambda_1, \dots, \lambda_n)$$

In general, singular values satisfy

$$\sigma_i = \sqrt{\lambda_i(A^T A)}$$

Therefore, as A is symmetric:

$$A^T A = A^2 = Q\Lambda Q^T Q\Lambda Q^T = Q\Lambda^2 Q^T$$

The eigenvalues of $A^T A$ are λ_i^2, so the singular values of A are

$$\sqrt{\lambda_i^2} = |\lambda_i|$$

Thus

$$\boxed{\{\sigma_1, \dots, \sigma_n\} = \{|\lambda_1|, \dots, |\lambda_n|\}}$$

The singular values are the absolute eigenvalues sorted in decreasing order.

2 More Facts About Singular Values

Recall the maximum- and minimum-gain definitions

$$\sigma_{\max}(X) = \max_{\|z\|=1} \|Xz\|, \qquad \sigma_{\min}(X) = \min_{\|z\|=1} \|Xz\|$$

(a) True.
Let e_i be the ith standard basis vector. The norm of row i of X is

$$\left(\sum_{j=1}^{n} |X_{ij}|^2 \right)^{1/2} = \|X^T e_i\|$$

Since $\|e_i\| = 1$ and X and X^T have the same singular values,

$$\|X^T e_i\| \leq \sigma_{\max}(X^T) = \sigma_{\max}(X)$$

Taking the maximum over i proves the statement.
(b) False.
If $X \in \mathbb{R}^{n\times n}$ is rank deficient, then some unit vector z satisfies $Xz = 0$, so

$$\sigma_{\min}(X) = 0$$

This can occur even when every row has positive norm. For example,

$$X = \begin{bmatrix} 1 & 0 \\ 1 & 0 \end{bmatrix}$$

Both row norms equal 1, but X has rank one, so the claimed inequality would give $0 \geq 1$.
(c) True.
We know that

$$\|XY\| \leq \|X\|\|Y\|$$

So for any z,

$$\|XYz\| \leq \sigma_{\max}(X)\|Yz\| \leq \sigma_{\max}(X)\sigma_{\max}(Y)\|z\|$$

Maximizing over all unit vectors z gives

$$\boxed{\sigma_{\max}(XY) \leq \sigma_{\max}(X)\sigma_{\max}(Y)}$$

(d) True.
For any z,

$$\|XYz\| \geq \sigma_{\min}(X)\|Yz\| \geq \sigma_{\min}(X)\sigma_{\min}(Y)\|z\|$$

Then, minimizing over all unit vectors z gives

$$\boxed{\sigma_{\min}(XY) \geq \sigma_{\min}(X)\sigma_{\min}(Y)}$$

(e) True.
For every unit vector z, (reverse) triangle inequality gives

$$\|(X+Y)z\| = \|Xz + Yz\| \geq \|Xz\| - \|Yz\| \geq \sigma_{\min}(X) - \sigma_{\max}(Y)$$

So taking the minimum over z gives

$$\boxed{\sigma_{\min}(X+Y) \geq \sigma_{\min}(X) - \sigma_{\max}(Y)}$$

3 Two Representations of an Ellipsoid

(a) Given $S = S^T > 0$,

$$x^T S x = \|S^{1/2}x\|_2^2$$

Therefore

$$\mathcal{E}_1 = \{x \mid x^T S x \leq 1\} = \{x \mid \|S^{1/2}x\|_2 \leq 1\}$$

Define $z = S^{1/2}x$. Since $S^{1/2}$ is invertible, $x = S^{-1/2}z$, and hence

$$\mathcal{E}_1 = \{S^{-1/2}z \mid \|z\|_2 \leq 1\}$$

Comparing with $\mathcal{E}_2$,

$$\boxed{A = S^{-1/2}}$$

(b)
Since $\det(A) \neq 0$, A is invertible. From $y = Ax$, we have $x = A^{-1}y$, so

$$\mathcal{E}_2 = \{y \mid \|A^{-1}y\|_2^2 \leq 1\} = \{y \mid y^T A^{-T} A^{-1} y \leq 1\}$$

Comparing with $\mathcal{E}_1$,

$$\boxed{S = A^{-T}A^{-1} = (AA^T)^{-1}}$$

This S is symmetric positive definite since, for $y \neq 0$,

$$y^T S y = \|A^{-1}y\|_2^2 > 0$$

(c)
Given S:
Equality of the two representations requires

$$AA^T = S^{-1}$$

Let

$$R = S^{1/2}A$$

Then $RR^T = I$, so R is orthogonal. Consequently, all possible matrices A are

$$\boxed{A = S^{-1/2}R, \qquad R^T R = RR^T = I}$$

Since R is orthogonal, it rotates or reflects the unit ball onto itself. Therefore $S^{-1/2}R$ produces the same ellipsoid for every orthogonal R.

Then, given A:
Conversely, every such A satisfies $AA^T = S^{-1}$, so this describes all choices.
For a fixed invertible A, the matrix S is unique:

$$\boxed{S = (AA^T)^{-1}}$$

4 Principal Component Analysis

Let
$$\widetilde{X} = \begin{bmatrix} \widetilde{x}_1 & \cdots & \widetilde{x}_N \end{bmatrix}, \qquad S = \frac{1}{N}\widetilde{X}\widetilde{X}^T$$

(a)

(i) The mean of the projected values $v^T x_i$ is $v^T \bar{x}$, so their centered values are $v^T \widetilde{x}_i$. Their sample variance is
$$\frac{1}{N}\sum_{i=1}^{N}(v^T\widetilde{x}_i)^2 = v^T\left(\frac{1}{N}\sum_{i=1}^{N}\widetilde{x}_i\widetilde{x}_i^T\right)v = v^TSv$$

(ii) The covariance matrix is symmetric. For every z,
$$z^TSz = \frac{1}{N}\|\widetilde{X}^Tz\|_2^2 \geq 0$$
so $S \succeq 0$. Using the full SVD
$$\widetilde{X} = U\Sigma V^T$$
gives
$$S = \frac{1}{N}U\Sigma\Sigma^TU^T$$
Therefore the left singular vectors u_i are eigenvectors of S, with eigenvalues
$$\boxed{\lambda_i = \frac{\sigma_i^2}{N}}$$

(iii) Expand a unit vector in the orthonormal eigenvector basis:
$$v = \sum_i \alpha_i u_i, \qquad \sum_i \alpha_i^2 = 1$$
Then
$$v^TSv = \sum_i \lambda_i\alpha_i^2 \leq \lambda_1$$
Equality is attained at $v = u_1$, so
$$\boxed{\max_{\|v\|_2=1} v^TSv = \lambda_1 = \frac{\sigma_1^2}{N}}$$

(b)

(i) Let $P = QQ^T$. Since
$$\widehat{x}_i = \bar{x} + P\widetilde{x}_i$$
we have
$$x_i - \widehat{x}_i = (I-P)\widetilde{x}_i$$
The vectors $P\widetilde{x}_i$ and $(I-P)\widetilde{x}_i$ are orthogonal, so the Pythagorean theorem gives
$$\|\widetilde{x}_i\|_2^2 = \|P\widetilde{x}_i\|_2^2 + \|(I-P)\widetilde{x}_i\|_2^2$$

Averaging over the data gives

$$V = \frac{1}{N}\sum_i \|\widetilde{x}_i\|_2^2 = \frac{1}{N}\sum_i \|QQ^T\widetilde{x}_i\|_2^2 + E(Q)$$

Also,

$$V = \frac{1}{N}\,\mathrm{trace}(\widetilde{X}\widetilde{X}^T) = \mathrm{trace}(S) = \sum_i \lambda_i$$

(ii) The variance captured by the projection is

$$\frac{1}{N}\sum_i \|QQ^T\widetilde{x}_i\|_2^2 = \mathrm{trace}(Q^TSQ)$$

Thus

$$E(Q) = \mathrm{trace}(S) - \mathrm{trace}(Q^TSQ)$$

The lecture result for symmetric matrices states that $\mathrm{trace}(Q^TSQ)$ is maximized by

$$Q = \begin{bmatrix} u_1 & \cdots & u_k \end{bmatrix}$$

with value $\lambda_1 + \cdots + \lambda_k$. Therefore

$$\boxed{E^\star = \sum_{i>k}\lambda_i = \frac{1}{N}\sum_{i>k}\sigma_i^2}$$

(c)
(i) The first 21 principal components are needed to capture at least 90% of the total variance.



(ii) The reconstructions below use the mean plus the projection onto the first k principal components. With $k = 2$, substantial digit-specific detail is lost. Around 6 components many digits are recognizable, and by 12 components the displayed digits are clear. Thus roughly 6–12 components are needed for recognizable reconstructions.



The mean image and first five principal components are shown below.



(iii) For $k = 10$, direct reconstruction gives

$$\frac{1}{N}\sum_{i=1}^{N} \|x_i - \widehat{x}_i\|_2^2 = 314.5149712423$$

The discarded singular values give

$$\frac{1}{N}\sum_{i>10} \sigma_i^2 = 314.5149712423$$

The absolute numerical difference is 2.27×10^{-13}. The fraction of total variance lost is

$$1 - \rho_{10} = 0.261773$$

so 26.18% of the variance is discarded.

4.1 Code

```
using LinearAlgebra
using Statistics
using Plots

include(joinpath(@__DIR__, "readclassjson.jl"))
data = readclassjson(joinpath(@__DIR__, "digits.json"))
X = data["X"]
n, N = size(X)

xbar = mean(X; dims=2)
Xc = X .- xbar
F = svd(Xc)

rho = cumsum(F.S .^ 2) ./ sum(F.S .^ 2)
k90 = findfirst(>=(0.90), rho)

plot(1:n, rho;
     xlabel="number of principal components k",
     ylabel="fraction of variance captured")
hline!([0.90]; linestyle=:dash)

function reconstruct(xc, k)
    Q = F.U[:, 1:k]
    return vec(xbar) + Q * (Q' * xc)
end

function digit_plot(x; title="")
    image = reshape(x, 8, 8)'
    return heatmap(image; color=:grays, yflip=true,
                   aspect_ratio=:equal, axis=false,
                   colorbar=false, title)
end

ks = [2, 6, 12, 24]
examples = [1, 2, 3]
panels = Any[]
for i in examples
    push!(panels, digit_plot(X[:, i]; title="original"))
    for k in ks
        xhat = reconstruct(Xc[:, i], k)
        push!(panels, digit_plot(xhat; title="k = $k"))
    end
end
plot(panels...; layout=(3, 5))

k = 10
Q = F.U[:, 1:k]
Xhat = xbar .+ Q * (Q' * Xc)
empirical_error = norm(X - Xhat)^2 / N
predicted_error = sum(F.S[(k + 1):end] .^ 2) / N
```

5 Smoothing

(a) The convolution matrix is

$$A_{ij} = \begin{cases} c_{j-i}, & -h \le j - i \le h \\ 0, & \text{otherwise} \end{cases}$$

where entries outside the signal are omitted. Then

$$y = Ax + w$$

For the data, $n = 200$ and $h = 20$. The singular values decay from

$$\sigma_{\max}(A) = 0.998929$$

to

$$\sigma_{\min}(A) = 2.40405 \times 10^{-9}$$



(b) The first six right singular vectors are smooth, oscillatory signals. The first has almost no oscillation, and each subsequent vector has a higher frequency. This is expected because convolution by the smooth kernel preserves low-frequency directions most strongly, so they correspond to the largest singular values.



Figure 1: First six right singular vectors

(c) The condition number is

$$\kappa(A) = \frac{\sigma_{\max}(A)}{\sigma_{\min}(A)} = 4.15520 \times 10^8$$

Writing $A = U\Sigma V^T$, the least-squares estimate contains terms

$$\frac{u_i^T y_{\text{meas}}}{\sigma_i} v_i$$

Thus noise in directions with very small σ_i is amplified enormously. The resulting error is

$$\|x - x_{\text{ls}}\|_2 = 2.34098 \times 10^8$$



Figure 2: Unregularized least-squares estimate

(d) For rank r, the truncated estimate is

$$x_{\text{est}}^{(r)} = V_r \Sigma_r^{-1} U_r^T y_{\text{meas}}$$

The observed errors for the requested ranks are

r	5	10	15	30	50
$\|x - x_{\text{est}}^{(r)}\|_2$	20.662	16.346	12.819	10.604	15.177

Small r gives an overly smooth estimate. Increasing r initially recovers the transitions in x, but at $r = 50$ more noise-dominated directions enter and the estimate becomes oscillatory.



(e) The error decreases as useful singular-vector components are added, reaches its minimum at $r = 28$, and then begins to increase as noise amplification outweighs the added signal information.



(f) Among $r = 1, \dots, 35$, the best value is

$$\boxed{r = 28}$$

with error

$$\|x - x_{\text{est}}^{(28)}\|_2 = 9.95897$$



(g) The Tikhonov objective is

$$J(x) = \|Ax - y\|^2 + \mu\|x\|^2$$

Setting its gradient to zero gives

$$(A^T A + \mu I)x_{\text{reg}} = A^T y_{\text{meas}}$$

and therefore

$$\boxed{x_{\text{reg}} = (A^T A + \mu I)^{-1} A^T y_{\text{meas}}}$$

There is a good estimate at

$$\mu = 0.05012$$

with error 9.95345.



(h) Let

$$A = U\Sigma V^T = \sum_{i=1}^{n} \sigma_i u_i v_i^T$$

Then

$$B = (A^T A + \mu I)^{-1} A^T = V \operatorname{diag}\left(\frac{\sigma_i}{\sigma_i^2 + \mu}\right) U^T$$

Define

$$g_i = \frac{\sigma_i}{\sigma_i^2 + \mu}$$

These gains are not necessarily sorted when the σ_i are sorted. Let π order them so that

$$g_{\pi(1)} \geq \cdots \geq g_{\pi(n)}$$

Then the SVD of B is obtained from

$$\boxed{\widetilde{\sigma}_j = g_{\pi(j)}, \qquad \widetilde{u}_j = v_{\pi(j)}, \qquad \widetilde{v}_j = u_{\pi(j)}}$$

(i) The norm of B is its largest singular value:

$$\boxed{\|B\|_2 = \max_i \frac{\sigma_i}{\sigma_i^2 + \mu}}$$

For the selected μ, this is approximately 2.23226.

(j) Using the SVD,

$$AB = U \operatorname{diag}\left(\frac{\sigma_i^2}{\sigma_i^2 + \mu}\right) U^T$$

so

$$AB - I = -U \operatorname{diag}\left(\frac{\mu}{\sigma_i^2 + \mu}\right) U^T$$

Therefore

$$\boxed{\max_{y \neq 0} \frac{\|ABy - y\|_2}{\|y\|_2} = \|AB - I\|_2 = \max_i \frac{\mu}{\sigma_i^2 + \mu} = \frac{\mu}{\sigma_{\min}(A)^2 + \mu}}$$

For the selected μ, this value is numerically 1.00000, since the smallest singular value is extremely close to zero.

5.1 Code

```
using LinearAlgebra
using Plots

include(joinpath(@__DIR__, "readclassjson.jl"))
data = readclassjson(joinpath(@__DIR__, "regl_data.json"))
n, c = data["n"], data["c"]
x, w = vec(data["x"]), vec(data["w"])

function convolution_matrix(c, n)
    h = div(length(c) - 1, 2)
    A = zeros(n, n)
    for i in 1:n, k in -h:h
        j = i + k
        if 1 <= j <= n
            A[i, j] = c[k + h + 1]
        end
    end
    return A
end

A = convolution_matrix(c, n)
ymeas = A * x + w
F = svd(A)

plot(1:n, F.S; yscale=:log10,
     xlabel="k", ylabel="sigma_k")
plot(1:n, F.V[:, 1:6]; layout=(3, 2))

xls = A \ ymeas
p_ls = plot(x; label="true x")
plot!(p_ls, xls; label="least-squares estimate")
```

```

function truncated_estimate(F, y, r)
    Ur = F.U[:, 1:r]
    Vr = F.V[:, 1:r]
    return Vr * ((Ur' * y) ./ F.S[1:r])
end

ranks = [5, 10, 15, 30, 50]
xtruncated = [
    truncated_estimate(F, ymeas, r) for r in ranks
]
rank_plots = map(zip(ranks, xtruncated)) do (r, xr)
    p = plot(x; label="true x")
    plot!(p, xr; label="estimate, r = $r")
end
plot(rank_plots...; layout=(3, 2))

tested_ranks = 1:35
rank_errors = [
    norm(x - truncated_estimate(F, ymeas, r))
    for r in tested_ranks
]
best_r = tested_ranks[argmin(rank_errors)]
plot(tested_ranks, rank_errors;
     xlabel="r", ylabel="norm(x - x_est)")
xbest = truncated_estimate(F, ymeas, best_r)
p_best = plot(x; label="true x")
plot!(p_best, xbest; label="best truncated estimate")

function tikhonov_estimate(F, y, mu)
    gains = F.S ./ (F.S .^ 2 .+ mu)
    return F.V * (gains .* (F.U' * y))
end

mus = 10.0 .^ range(-10, 0; length=301)
reg_errors = [
    norm(x - tikhonov_estimate(F, ymeas, mu))
    for mu in mus
]
best_mu = mus[argmin(reg_errors)]
xreg = tikhonov_estimate(F, ymeas, best_mu)
p_reg = plot(x; label="true x")
plot!(p_reg, xreg; label="Tikhonov estimate")
```